THE DIFFERENTIALGEOMETRY SOFTWARE PROJECT

How To Create A Lie Algebra

Synopsis

• We show how to create a Lie algebra in Maple using three of the most common approaches: matrices, vector fields and structure equations.

Examples

Load in the required packages.

```
> with(DifferentialGeometry): with(LieAlgebras):
```

Example 1.

The well-known Pauli matrices (upon multiplication by I) are a set of 3 anti-Hermitian matrics which define a real 3-dimensional matrix algebra. Here are the Pauli matrices:

```
> A := [Matrix([[0,I], [I,0]]),  Matrix([[0, 1], [-1, 0]]), Matrix([[I, 0], [0, -I]])];
```

$$A := \left[\begin{bmatrix} 0 & I \\ I & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 \\ -1 & 0 \end{bmatrix}, \begin{bmatrix} I & 0 \\ 0 & -I \end{bmatrix} \right] \tag{1}$$

The Lie algebra bracket is the matrix commutator. The output of the command LieAlgebraData is a list of the non-zero brackets, where by default $e1, e2, e3$ denote the 1st, 2nd and 3rd Pauli matrices.

```
> LD1 := LieAlgebraData(A, Alg1);
```

$$LD1 := [[e1, e2] = -2\, e3, [e1, e3] = 2\, e2, [e2, e3] = -2\, e1] \tag{2}$$

We use the command DGsetup to store these structure equations in memory.

```
> DGsetup(LD1);
```

$$\textit{Lie algebra: Alg1} \tag{3}$$

At this point one can now invoke many of the commands in the LieAlgebras package. For example, here is the Killing matrix for this Lie algebra:

```
Alg1 > Killing(Alg1);
```

(4)

$$\begin{bmatrix} -8 & 0 & 0 \\ 0 & -8 & 0 \\ 0 & 0 & -8 \end{bmatrix} \quad (4)$$

Example 2.

The infinitesimal generators for translations and rotations in the *xy* plane are vector fields which define a 3-dimensional Lie algebra. The bracket operation is the Lie bracket of vector fields. We create a coordinate system for the *xy* plane, define the vector fields and compute their brackets.

```
> DGsetup([x, y], R2);
```

frame name: R2 **(5)**

```
R2 > EuclideanGenerators  := [D_x,  D_y, x*D_y - y*D_x];
```

$EuclideanGenerators := [D\_x, D\_y, x\, D\_y - y\, D\_x]$ **(6)**

```
R2 > LD2 := LieAlgebraData(EuclideanGenerators, Alg2);
```

$LD2 := [[e1, e3] = e2, [e2, e3] = -e1]$ **(7)**

The labels of the basis of the Lie algebra can be specified when the Lie algebra is initialized with *DGsetup*. Here we use *X*, *Y*, *R* as labels for the 1st, 2nd and 3rd vectors in the Lie algebra and α, β, θ as the labels for the dual 1-forms.

```
R2 > DGsetup(LD2, [X, Y, R], [alpha, beta, theta]);
```

Lie algebra: Alg2 **(8)**

Here is the multiplication table - the table of brackets - for the Lie algebra.

```
Alg2 > MultiplicationTable("LieTable");
```

	X	Y	R
X	0	0	Y
Y	0	0	-X
R	-Y	X	0

(9)

We can use the Query command to check that this Lie algebra is solvable.

```
Alg2 > Query("Solvable");
```

true **(10)**

Example 3.

An abstract Lie algebra can always be created by specifying the non-zero Lie brackets. In the following $[x1, x2, x3, x4, x5]$ are unassigned names which denote the basis elements for the Lie algebra, which is defined by the following brackets.

```
Alg2 > StrEq := [[x2, x3] = x1, [x2, x5] = x3, [x4, x5] = x4];
```
$$StrEq := [[x2, x3] = x1, [x2, x5] = x3, [x4, x5] = x4] \quad (11)$$

We eonvert the brackets to a Maple Lie algebra data struture with *LieAlgebraData* and initialize with *DGsetup*.

```
> LD3 := LieAlgebraData(StrEq, [x1, x2, x3, x4, x5], Alg3);
```
$$LD3 := [[e2, e3] = e1, [e2, e5] = e3, [e4, e5] = e4] \quad (12)$$

```
Alg2 > DGsetup(LD3);
```
Lie algebra: Alg3 **(13)**

The command Derivations calculates the Lie algebra of infinitesimal automorphism of a Lie algebra. It is just one of many ways to create a new Lie algebra from a given one.

```
Alg3 > Derivations(Alg3, "Full");
```
$$\left[\begin{bmatrix}1&0&0&0&0\\0&\frac{1}{2}&0&0&0\\0&0&\frac{1}{2}&0&0\\0&0&0&0&0\\0&0&0&0&0\end{bmatrix}, \begin{bmatrix}0&1&0&0&0\\0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&0\end{bmatrix}, \begin{bmatrix}0&0&0&0&0\\0&0&0&0&0\\0&1&0&0&0\\0&0&0&0&0\\0&0&0&0&0\end{bmatrix}, \begin{bmatrix}0&0&1&0&0\\0&0&0&0&0\\0&0&0&0&1\\0&0&0&0&0\\0&0&0&0&0\end{bmatrix}, \begin{bmatrix}0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&0\\0&0&0&1&0\\0&0&0&0&0\end{bmatrix}, \begin{bmatrix}0&0&0&0&1\\0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&0\end{bmatrix}, \begin{bmatrix}0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&0\\0&0&0&0&1\\0&0&0&0&0\end{bmatrix}\right] \quad (14)$$

Commands Illustrated

- LieAlgebras, Derivations, DGsetup, LieAlgebraData, Killing, MultiplicationTable, Query

Related Commands

- The following commands provide additional ways to create Lie algebras: Derivations, InfinitesimalHolonomy, InfinitesimalSymmetriesOfGeometricObjectFields, KillingVectors, LieGroup, SimpleLieAlgebraData, StandardRepresentation, SymbolAlgebra

References

- M. L. Curtis, *Matrix Algebras*
- W. Fulton, J. Harris, *Representation Theory - A First Course*
- P. J. Olver, *Applications of Lie Groups to Differential Equations*

• D. H. Sattinger, O. L. Weaver, *Lie Groups and Algebras with Applications to Physics, Geometry, and Mechanics*.
• http://en.wikipedia.org/wiki/Lie_algebra

Release Notes

• The illustrated commands are available in Maple 11 and subsequent releases.

Author

Ian M. Anderson,
Department of Mathematics and Statistics , Utah State University
July 4, 2012